# Copyright 2015 Google Inc. All Rights Reserved.
#
# Licensed under the Apache License, Version 2.0 (the "License");
# you may not use this file except in compliance with the License.
# You may obtain a copy of the License at
#
#     http://www.apache.org/licenses/LICENSE-2.0
#
# Unless required by applicable law or agreed to in writing, software
# distributed under the License is distributed on an "AS IS" BASIS,
# WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
# See the License for the specific language governing permissions and
# limitations under the License.

"""Build definitions for Closure templates. The template targets may be used in
closure_js_binary rules.

Example:

  closure_template_library(
      name = "hello_soy",
      srcs = ["hello.soy"],
  )

This rule will produce hello_soy.js.
"""

_SOY_FILE_TYPE = FileType([".soy"])

def _impl(ctx):
  srcs = ctx.files.srcs
  args = [
      "--cssHandlingScheme",
      "goog",
      "--outputPathFormat",
      ctx.outputs.out.path,
      "--shouldGenerateJsdoc",
      "--shouldProvideRequireSoyNamespaces",
      "--srcs",
      ",".join([src.path for src in srcs])
  ]

  ctx.action(
      inputs=list(srcs),
      outputs=[ctx.outputs.out],
      arguments=args,
      executable=ctx.executable._closure_templates)

  return struct(
      files=set(), transitive_js_externs=set(),
      transitive_js_srcs=set([ctx.outputs.out]))

closure_template_library = rule(
    implementation=_impl,
    attrs={
        "srcs": attr.label_list(allow_files=_SOY_FILE_TYPE),
        "_closure_templates": attr.label(
            default=Label("//external:closure_templates_"),
            executable=True),
    },
    outputs={"out": "%{name}.js"})
